

Übersichtliche Projektplanung erstellen

Einfache Kosten- und Terminüberwachung in Excel

Teil 1: Der Projektplan



Ignatz Schels

IT-Experte, DV-Dozent und Journalist

Geht es um Kosten und Termine, ziehen viele Projektarbeiter die Tabellenkalkulation Microsoft Excel klassischen PM-Werkzeugen wie MS Project vor. Excel kann zwar nicht alles, was ausgereifte PM-Systeme leisten, hinsichtlich seiner Flexibilität ist es ihnen aber überlegen. Mit Excel lassen sich besonders einfach Projekt- und Terminpläne anlegen, da das Programm mit besseren Kalkulations- und Analysewerkzeugen und optimalen Visualisierungen über Diagramme punktet.

Dieser zweiteilige Beitrag zeigt, wie Sie Excel für die Erfassung, Planung und Fortschrittskontrolle kleiner bis mittelgroßer Projekte verwenden können. In diesem ersten Teil der Artikelfolge erfahren Sie, wie Sie einen übersichtlichen Projektplan in Excel erzeugen, der die Basis für die spätere Kosten- und Terminüberwachung bildet. Der zweite Teil des Beitrags beschreibt, wie Sie aus den Daten des Projektplans eine Meilensteintrend- und eine Kostentrend-Analyse erstellen. Grundlegende Kenntnisse über Zellbezüge, Formatierungen und einfache Kalkulationsfunktionen sollten Sie hierfür mitbringen.



Um die einzelnen Schritte nachzuvollziehen, nutzen Sie die Arbeitshilfe *Projekt_Magic_AirStar.xlsx*, die beim Artikel als Download zur Verfügung steht.

Das Beispielprojekt: Produktentwicklung "Händetrockner"

Sie sind technischer Leiter einer Firma, die elektronische Geräte für den Haushalts- und Sanitärbereich herstellt. Die Produktpalette reicht vom Toaster über Staubsauger bis hin zur Klimaanlage. Marketing und Sales melden sinkende Verkaufszahlen im Hygienebereich, vor allem das Produktsegment "Händetrockner" ist davon betroffen, da die Geräte von Design und Konzept her veraltet sind.

Ihre Aufgabe als Projektleiter besteht darin, einen neuen, innovativen Händetrockner zu entwickeln. Die Spezifikationen sind festgelegt, der Endtermin steht fest (zwei Wochen vor der internationalen Fachmesse für Sanitärausstattung). Sie haben Ihr Projektteam aus Ingenieuren, Designern und Technikern zusammengestellt und nach einigen intensiven Projektsitzungen ist auch der vorläufige Projektplan erstellt.

Den Excel-Projektplan anlegen

Wie in der klassischen Projektmanagement-Software wird auch in Excel zunächst eine Liste mit den Projektschritten (Vorgängen) angelegt. Die Arbeitshilfe enthält im Blatt "Erstplanung" diese Liste mit einer Zuordnung der einzelnen Arbeitspakete zu den jeweiligen Phasen (siehe Bild 1).

Um diese Auflistung in einen Projektplan umzuwandeln, gehen Sie wie folgt vor:

- Kopieren Sie zunächst das Tabellenblatt, indem Sie das Register mit gedrückter Maustaste nach rechts ziehen. Klicken Sie doppelt in das Register und benennen Sie es "Projektplan".
- Fügen Sie eine neue Spalte A ein und tragen Sie die Überschriften für die Liste ein:
A3: ID
A4: Vorgang
- Kennzeichnen Sie in Spalte A alle Einträge, geben Sie den Meilensteinen die ID "M" und den Phasenbezeichnungen die ID "P".

Entfernen Sie anschließend alle Schrift- und Zellformatierungen aus der Projektliste. Mit einer bedingten Formatierung werden Meilensteine, Phasen und Arbeitspakete mit jeweils unterschiedlichen Schriftfarben gekennzeichnet (siehe Bild 2):

- Markieren Sie alle Zeilen des Projektplans ab Zeile 4. Ziehen Sie den Mauszeiger dazu mit gedrückter Maustaste über die Zeilennummern.
- Wählen Sie Start / Format / Bedingte Formatierung / Neue Regel. Wählen Sie Formel zur Ermittlung der zu formatierenden Zellen verwenden. Tragen Sie hier diese Formel ein: `=A3="M"`

	A	B
1	Projekt "Magic AirStar"	
2		
3	Projektbeginn	
4	Design	← Phase
5	Produktplanungsbesprechung durchführen	
6	Skizzen erstellen	
7	Produktdesign festlegen	← Vorgänge
8	Marketingstrategie festlegen	
9	Technische Zeichnungen erstellen	
10	Abschluss Designphase	← Meilenstein
11	Gehäuse	
12	Material auswählen	
13	Gehäuse zeichnen	
14	Gittermodell erstellen	
15	Muster herstellen	
16	Abnahme Gehäuse	
17	Elektronik	
18	Materialbeschaffung	
19	Leiterplatten entwickeln	
20	Elektronik fertigen	
21	Abnahme Elektronik	
22	Motor	
23	Lieferantenauswahl	
24	Motor entwickeln	
25	Motor fertigen	
26	Abnahme Motor	
27	Testphase	
28	Funktionsmuster herstellen	
29	Funktionsmuster testen	
30	Werkzeug vorbereiten	
31	Nacharbeit	
32	Abschluß Testphase	
33	Nullserienfertigung	
34	Nullseriebesprechung	
35	Nullserie produzieren	
36	Nullserie fertiggestellt	
37	Projektende	
38		
39		
40		
41		

Erstplanung

Bild 1: Der grobe Projektplan beinhaltet Phasen, Vorgänge und Meilensteine

- Wählen Sie unter Formatieren für Meilensteine die Schriftfarbe Magenta. Bestätigen Sie zweimal mit OK.
- Starten Sie die bedingte Formatierung erneut und stellen Sie eine zweite Formel-Regel auf. Die Formel lautet: `= $A4="P"`
- Für Phasen wählen Sie die Schriftfarbe Grün.

Achten Sie auf das "\$"-Zeichen vor der Spaltenbezeichnung, es stellt sicher, dass das Format auf die ganze Zeile angewandt wird.

Unter *Start / Formatvorlagen / Bedingte Formatierung / Regeln verwalten* finden Sie eine Übersicht über die angelegten Regeln. Hier können Sie bei Bedarf auch Regeln ändern oder zu formatierende Bereiche neu festlegen.




Bild 2: Projektplan mit IDs für Phasen und Meilensteine und bedingter Formatierung

Die Terminplanung erstellen

Die nächsten Spalten des Projektplans werden mit "Beginn", "Dauer" und "Ende" überschrieben. Nachdem Sie Vorgangsbeginn und Vorgangsdauer in Tagen eingetragen haben, errechnet Excel automatisch das Vorgangsende. Geben Sie in Zelle E4 das Datum des Projektbeginns ein: 3.1.2017

Datenüberprüfungen zur Sicherung Ihrer Eingaben

Da in den Datumsspalten nur Datumswerte und in der Dauer-Spalte nur Zahlen erlaubt sind, verwenden Sie je eine Datenüberprüfung, um die Eingaben abzusichern:

- Markieren Sie die Spalte C und wählen Sie Daten / Datentools / Datenüberprüfung.
- Schalten Sie unter Gültigkeitskriterien / Zulassen auf Datum um. Geben Sie diese Voreinstellungen ein:
Daten: Größer oder gleich

Anfangsdatum: =\$E\$4

- Tragen Sie im Register Fehlermeldung ein:

Typ: Stopp

Titel: Achtung! Falsches Datum

Fehlermeldung: Datum darf nicht vor dem Projektbeginn liegen!

Damit sind in Spalte C nur noch Datumswerte in den Spalten erlaubt, die nach dem Projektbeginn liegen – die Fehlermeldung weist bei Falscheingaben darauf hin. Entfernen Sie die Datenüberprüfung aus der Zelle E4, wenn Sie den Projektbeginn ändern wollen.

Sichern Sie zudem Spalte D (Dauer) mit einer Datenüberprüfung ab, die Dezimalzahlen zulässt. Wenn Sie im Projektplan halbe Tage vorsehen, verwenden Sie den Typ *Dezimal* mit einem Minimum von 0,5, wenn Sie in der Eingabe nur ganze Tage erlauben wollen, wählen Sie den Typ *Ganze Zahl* (Tabelle 1). In unserem Beispiel sind ganze Tage angegeben.

Vorgangsdauer mit halben Tagen	Vorgangsdauer mit ganzen Tagen
Zulassen: Dezimal Daten: Größer als Minimum: 0,5 Fehlermeldung: Achtung! Bitte Projektdauer in Tagen (z. B. 1 oder 1,5) eingeben!	Zulassen: Ganze Zahl Daten: Größer als Minimum: 0 Fehlermeldung: Achtung! Bitte Projektdauer in ganzen Tagen eingeben!

Tabelle 1: Bei der Projektdauer können Sie mit halben oder ganzen Tagen planen

Zahlenformate für Projekttermine festlegen

Das Zahlenformat der Spalte C formatiert das Datum des Projektbeginns mit Angabe des Wochentags. Markieren Sie Spalte C und wählen Sie im Kontextmenü der rechten Maustaste *Zellen formatieren*. Schalten Sie auf *Benutzerdefiniert* um und geben Sie unter *Typ* den Zahlenformatcode ein, der die abgekürzte Tagesbezeichnung (M, Di, Mi ...) linksbündig und das Datum rechtsbündig setzt (Die Kombination * <Leertaste> sorgt für diese Füllung): TTT* TT.MM.JJ

Die Spalte D mit der Projektdauer kann wahlweise mit einem Textzusatz für den Tag oder als reine Dezimalzahl formatiert werden. In unserem Beispiel wurde der Textzusatz "t" gewählt.

Vorgangsdauer mit halben Tagen	Vorgangsdauer mit ganzen Tagen
0,0 oder 0,0" t" oder 0,0" Tage"	0 oder 0" t" oder 0" Tage"

Tabelle 2: Die Projektdauer können Sie als reine Zahlen oder mit einem Textzusatz angeben

Endtermine für Vorgänge berechnen

Wenn die Vorgangsdauer in Kalendertagen eingegeben wird, lässt sich der Endtermin einfach durch Addition der eingegebenen Tage und dem Vorgangsbeginn berechnen. Geben Sie dazu diese Formel in Zelle E6 ein:

=WENN(UND(C6<>"";D6<>"") ;C6+D6;"")

Kopieren Sie die Formel mit dem Füllkästchen unten am Zellzeiger bis zum Ende des Projektplans zum letzten Meilenstein "Projektende". Geben Sie jetzt die geplanten Termine und die Dauer der einzelnen Vorgänge ein (Bild 3).

E6		=WENN(UND(C6<>"";D6<>"");C6+D6;"")		
A	B	C	D	E
1	Projekt "Magic AirStar"			
2				
3	ID Vorgang	Beginn	Dauer	Ende
4	M Projektbeginn			Di 03.01.17
5	P Design			
6	Produktplanungsbesprechung durchführen	Di 03.01.17	12 t	So 15.01.17
7	Skizzen erstellen	Di 10.01.17	7 t	Di 17.01.17
8	Produktdesign festlegen	Di 17.01.17	9 t	Do 26.01.17
9	Marketingstrategie festlegen	Do 26.01.17	11 t	Mo 06.02.17
10	Technische Zeichnungen erstellen	Mo 06.02.17	10 t	Do 16.02.17
11	M Abschluss Designphase			
12	P Gehäuse			
13	Material auswählen	Do 16.02.17	5 t	Di 21.02.17
14	Gehäuse zeichnen	Mi 22.02.17	12 t	Mo 06.03.17
15	Gittermodell erstellen	Mo 06.03.17	4 t	Fr 10.03.17
16	Muster herstellen	Mo 13.03.17	8 t	Di 21.03.17
17	M Abnahme Gehäuse			
18	P Elektronik	Do 16.02.17	12 t	Di 28.02.17
19	Materialbeschaffung	Di 28.02.17	6 t	Mo 06.03.17
20	Leiterplatten entwickeln	Mo 06.03.17	14 t	Mo 20.03.17
21	Elektronik fertigen	So 12.03.17	14 t	So 26.03.17
22	M Abnahme Elektronik			
23	P Motor			

Bild 3: Das Vorgangsende berechnet sich aus der Summe von Beginn und Dauer

Termine der Meilensteine berechnen

Die Entermine der einzelnen Phasen werden im Projektplan als Meilensteine aufgeführt. Tragen Sie in den Zeilen mit der ID "M" die Formel ein, die den größten Datumswert der Phase berechnet:

E11: =MAX(E6:E10)

E17:=MAX(E13:E16)

E22: =MAX(E18:E21)

E27: =MAX(E24:E26)

E33: =MAX(E29:E32)

E37: =MAX(E35:E36)

E38: =MAX(E4:E37)

Projekttag berechnen

Um die Projektkosten realistisch planen und überwachen zu können, sollten auch die Nettotage, d.h. die Differenzen zwischen Beginn- und Enddatum abzüglich der Wochenendtage und Feiertage, berechnet werden. Im Tabellenblatt "Feiertage" finden Sie eine Feiertagsberechnung mit automatischer Erkennung des Projektjahrs. Die Liste enthält die Feiertage für zwei Jahre, damit Projektvorgänge, die über das Jahresende dauern, berücksichtigt werden. Kreuzen Sie in Spalte B die Feiertage an, die für Ihr Bundesland zutreffen, geben Sie dazu einfach ein "x" ein. Tragen Sie in die Zelle A1 die Jahreszahl des Projektbeginns ein oder berechnen Sie diese einfach aus der Zelle mit dem Datum des Projektbeginns: =Jahr(PSP!\$A\$4).

Hinweis: Wie Sie mit Excel Feiertage bundeslandspezifisch berechnen, beschreibt der Artikel "Feiertage berechnen mit Microsoft Excel" in Ausgabe 12/2006.

Die Datumswerte der Feiertagsliste in Spalte C sind mit dem Bereichsnamen "Feiertage" versehen. Kontrollieren Sie diesen über *Formeln / Definierte Namen / Namens-Manager*. Geben Sie bei Bedarf weitere Datumswerte für zusätzliche freie Projektstage ein und ändern Sie den Zellbezug des Bereichsnamens "Feiertage".

Berechnen Sie im Projektplan in Spalte F mit der Überschrift "Projektstage" die Anzahl der Tage, die zwischen dem Enddatum und dem Startdatum des ersten Vorgangs liegen und nicht auf einen Wochenendtag (Samstag, Sonntag) oder Feiertag fallen. Verwenden Sie dazu die Funktion NETTOARBEITSTAGE():

F6: =WENN(UND(C6<>"";D6<>"");NETTOARBEITSTAGE(C6;E6;Feiertage);"")

Die Funktion NETTOARBEITSTAGE.INTL() bietet die Möglichkeit, die Tage frei zu definieren, die im Projekt nicht als Projektstage gezählt werden. Geben Sie im letzten Argument eine Ziffer für die passende Kombination an (wird nach Eingabe des Semikolons angezeigt):



Kopieren Sie die Formel bis zum Ende des Projektplans. Die berechneten Tage werden nur in Vorgangszellen angezeigt.

=NETTOARBEITSTAGE.INTL(Beginn;Ende;[Wochenende];[Feiertage])

Bisher haben Sie der ursprünglichen Rohfassung des Projektplans Beginn, Dauer und Ende hinzugefügt sowie die Projektstage berechnet. In den nächsten Schritten erfahren Sie, wie Sie einen Projektplan gliedern und die Projektkosten berechnen.

	A	B	C	D	E	F	G
1	2017						
2		x	So 01. Januar	Neujahrstag			
3		x	Fr 06. Januar	Hl. Drei Könige			
4				Rosenmontag			
5				Faschingsdienstag			
6		x	Fr 14. April	Karfreitag			
7		x	So 16. April	Ostersonntag			
8		x	Mo 17. April	Ostermontag			
9		x	Mo 01. Mai	Tag der Arbeit			
10		x	Do 25. Mai	Christi Himmelfahrt			
11		x	So 04. Juni	Pfingstsonntag			
12		x	Mo 05. Juni	Pfingstmontag			
13		x	Do 15. Juni	Fronleichnam			
14				Augsburger Friedensfest			
15				Mariä Himmelfahrt			
16		x	Di 03. Oktober	Tag der d. Einheit			
17				Reformationstag			
18		x	Mi 01. November	Allerheiligen			

Bild 4: Die automatische Feiertagsliste wird für die Berechnung der Projektstage verwendet

F6								
=WENN(UND(C6<>"";D6<>"");NETTOARBEITSTAGE.INTL(C6;E6;1;Feiertage);"")								
A	B	C	D	E	F	G	H	I
ID	Vorgang	Beginn	Dauer	Ende	Projekttag			
M	Projektbeginn			Di 03.01.17				
P	Design							
6	Produktplanungsbesprechung durchführen	Di 03.01.17	12 t	So 15.01.17	8 t			
7	Skizzen erstellen	Di 10.01.17	7 t	Di 17.01.17	6 t			
8	Produktdesign festlegen	Di 17.01.17	9 t	Do 26.01.17	8 t			
9	Marketingstrategie festlegen	Do 26.01.17	11 t	Mo 06.02.17	8 t			
10	Technische Zeichnungen erstellen	Mo 06.02.17	10 t	Do 16.02.17	9 t			
M	Abschluss Designphase			Do 16.02.17				
P	Gehäuse							
13	Material auswählen	Do 16.02.17	5 t	Di 21.02.17	4 t			
14	Gehäuse zeichnen	Mi 22.02.17	12 t	Mo 06.03.17	9 t			
15	Gittermodell erstellen	Mo 06.03.17	4 t	Fr 10.03.17	5 t			
16	Muster herstellen	Mo 13.03.17	8 t	Di 21.03.17	7 t			
M	Abnahme Gehäuse			Di 21.03.17				
P	Elektronik	Do 16.02.17	12 t	Di 28.02.17	9 t			

Bild 5: Mit der Funktion NETTOARBEITSTAGE() oder NETTOARBEITSTAGE.INTL() können die Projekttag berechnet werden

Projektplan gliedern

Große Projektpläne werden schnell unüberschaubar und zwingen zum ständigen, zeitraubenden Blättern in Zeilen und Spalten der Excel-Datei. Arbeiten Sie deshalb mit Gliederungselementen, um die Übersicht zu behalten. Gliedern Sie die Vorgänge so, dass sich die Arbeitspakete jeder Phase geschlossen ein- und ausblenden lassen (Bild 6):

- Klicken Sie in der Gruppe Daten / Gliederung auf das Dialogfeldkästchen rechts unten und entfernen Sie in den Einstellungen das Häkchen neben der Option Hauptzeilen unter Detaildaten.
- Markieren Sie die Vorgänge und den Meilenstein der ersten Phase (die Phasenbezeichnung selbst wird nicht markiert) und wählen Sie Daten / Gliederung / Gruppieren. Achten Sie darauf, dass immer die gesamte Zeile markiert sein muss, andernfalls fragt Excel nach, ob Sie nach Zeilen oder Spalten gliedern möchten.
- Verfahren Sie für die restlichen Phasen entsprechend: Wiederholen Sie den letzten Vorgang mit der Funktionstaste F4.

Durch die Untergliederung erhalten Sie am linken Rand Gliederungssymbole. Mit einem Klick auf die Ebenen-Nummer 1 blenden Sie alle Vorgänge aus und zeigen nur noch die Phasenbegriffe an. Klicken Sie auf die Ebenen-Nummer 2, wird alles wieder eingeblendet.

Zur Spaltengliederung markieren Sie die Spalten "Beginn", "Dauer" und "Ende". Mit Daten / Gliederung / Gruppieren setzen Sie diese eine Ebene tiefer. Die Einstellung *Hauptspalten rechts von Detaildaten* sorgt dafür, dass die Projekttagberechnung als Ergebnisspalte rechts von den gegliederten Spalten verwendet wird.

Klicken Sie auf ein Gliederungssymbol im Zeilenbereich, um eine Phase aus- (Minuszeichen) oder einzublenden (Pluszeichen) und blenden Sie die Terminalsäulen per Klick auf das Minuszeichen im Spaltenkopf ein und wieder aus. Wenn Sie noch die Summen der Projekttag berechnen, können Sie auch größere Projektpläne schnell auf die wichtigsten Eckdaten reduzieren.




Bild 6: Gliederungssymbole machen große Projektpläne besser lesbar

Projektbudget und -kosten berechnen

So einfach und komfortabel wie die Terminplanung gestaltet sich in Excel auch die Kostenplanung. Für eine realistische Kostenüberwachung legen Sie das Projektbudget in einem separaten Blatt fest. Reservieren Sie die nächsten Spalten des Projektstrukturplans für die Budgetierung, tragen Sie in die Zellen G3 bis K3 (in unserem Beispiel im Tabellenblatt "Projektplan gegliedert") folgende Spaltenüberschriften ein:

G3: Plankosten

H3: Istkosten

I3: Restkosten

J3: Sollkosten

K3: Ampel

Diese Spalten können Sie wieder per Gliederungsfunktion in eine zweite Ebene befördern. Verwenden Sie die Ampelspalte als Ergebnisspalte und gliedern Sie die Spalten G:J mit *Daten / Gliederung / Gruppieren* in die nächste Spaltenebene.

Plankosten festlegen

Die Plankosten sind die Kosten, die bei der Projektplanung prognostiziert werden. Sie werden in der Regel pro Arbeitspaket fixiert, können sich aber auch auf eine Phase beziehen, wenn der Detaillierungsgrad nicht hoch genug ist. In der Praxis wird sich der Plankostenwert für ein Arbeitspaket oder eine Projektphase aus mehreren Faktoren zusammensetzen, z.B. aus Personal- und Materialkosten. In diesem Fall ist eine separate Tabelle erforderlich, in der die Kosten berechnet werden. In unserem Beispiel wollen wir die Plankosten für die Projektphasen kalkulieren. Gehen Sie dazu wie folgt vor (Bild 7):

- Markieren Sie den kompletten Projektplan und schalten Sie mit *Daten / Sortieren und Filtern / Filtern* die Filterpfeile ein.
- Filtern Sie in der ersten Spalte nach der ID "P" und kopieren Sie die Phasen in Spalte B.
- Legen Sie ein neues Tabellenblatt an und fügen Sie die kopierten Phasen ab Zelle A2 ein.
- Tragen Sie in den nachfolgenden Spalten die Kosten ein, trennen Sie diese in Personalkosten, Materialkosten und sonstige Kosten auf. Summieren Sie die Kosten zeilenweise mit der Funktion

SUMME(), kopieren Sie die Formel nach unten bis zum Ende des Kostenplans. Für die Zeile E2 geben Sie ein: E2: =SUMME(B2:D2)

- Markieren Sie den Bereich A1:E7 und weisen Sie ihm über *Formeln / Definierte Namen / Namen definieren* den Bereichsnamen "Kostenplan" zu. Sie können den Bereichsnamen auch in das Namensfeld links oben schreiben. Drücken Sie zum Abschluss die Eingabetaste.

Plankosten mit Projektplan verknüpfen

Mit den Phasenbezeichnungen bietet unsere Planung die Möglichkeit, die Kosten per Verweis aus dem Kostenplan in den Projektstrukturplan zu verknüpfen.

Schalten Sie dazu zurück auf den Projektplan und geben Sie in der ersten Projektzeile die Formel zur Berechnung der Plankosten für die Phase ein:

	A	B	C	D	E
1		Personalkosten	Materialkosten	Sonstige Kosten	Summe
2	Design	12.000	4.000	2.300	18.300
3	Gehäuse	12.000	16.000	5.000	33.000
4	Elektronik	21.000	21.000	6.000	48.000
5	Motor	15.000	15.000	5.000	35.000
6	Testphase	12.000	2.000	1.000	15.000
7	Nullserienfertigung	12.000	42.000	8.000	62.000
8					
9					

Bild 7: Der Projektkostenplan wird für die Projektphasen erstellt

G3: Plankosten

G5: =WENN(A5="P";SVERWEIS(B5;Kostenplan;5;FALSCH);"")

Die Formel prüft zunächst mit der WENN-Funktion, ob es sich bei der Zeile um eine Phase handelt. Ist das der Fall, sucht die Funktion SVERWEIS im Kostenplan nach dem Phasenbegriff und liefert als Ergebnis die Kostensumme aus Spalte 5. Das letzte Argument FALSCH ist erforderlich, damit diese Suche eindeutig ausfällt und nicht der nächst kleinere Wert akzeptiert wird.

Kopieren Sie die Formel anschließend nach unten bis zur letzten Zeile im Projektstrukturplan (siehe Bild 8).

Istkosten, Restkosten und Sollkosten

Als "Istkosten" wird der Betrag bezeichnet, der für das Arbeitspaket oder die Phase bereits angefallen ist. Diese Kosten werden im Projektverlauf eingetragen. Die Istkosten lassen sich ebenfalls aus einem Kostenplan verknüpfen, was den Vorteil bietet, dass sich die Zahlen dort detailliert darstellen lassen. So werden z.B. Arbeitsstunden auf Phasen oder Arbeitspakete umgelegt oder Kosten für Zeitarbeit, Stundenlöhne u.a. mit einberechnet.

Restkosten sind die Kosten, die noch für die Fertigstellung des Arbeitspakets erforderlich sein werden. Auch diese Kosten werden im Projektverlauf erfasst bzw. zum Berichtszeitpunkt eingetragen.

Die Sollkosten errechnen sich aus der Addition der Istkosten und der Restkosten. Schreiben Sie folgende Formel in der Sollkostenspalte in die Zeile der ersten Phase:

J5: =WENN(UND(H5<>"";I5<>"");I5+H5;"")

Kopieren Sie die Formel anschließend nach unten auf die übrigen Zeilen im Projektstrukturplan.

Ampelformatierung für Soll/Ist-Vergleich

In der nächsten Spalte des Projektplans wird die Kostenlage der einzelnen Vorgänge optisch visualisiert. Eine Ampelformatierung mit Smileys meldet, ob die Summe der Istkosten und der zu erwartenden Restkosten die Plankosten übersteigt (rot), ob die Kosten im Rahmen sind (gelb) oder ob zu erwarten ist, dass der Kostenrahmen deutlich unterschritten wird (grün). Erstellen Sie mit einer Kombination aus WENN-Funktion und bedingter Formatierung eine Ampelfunktion, die den Status der Abweichung visualisiert. Gehen Sie dazu wie folgt vor:

- Schreiben Sie nachfolgende Formel in die Zeile der ersten Phase. Die Formel prüft, ob der Sollkostenwert mehr als 1.000 (Euro) über bzw. unter dem Planwert liegt und trägt ein entsprechendes Zeichen ein:
K5: =WENN(UND(G5<>"";J5<>"");WENN(UND(J5<=G5+1000;J5>=G5-1000);"K";WENN(J5>G5+1000;"L";"J"));"")
- Formatieren Sie die Spalte K ab K5 über Start / Schriftart mit der Schriftart "WingDings", um die den Buchstaben J, K und L zugeordneten Smileys anzuzeigen.
- Erstellen Sie unter Start / Formatvorlagen / Bedingte Formatierung je eine Bedingungsregel für die drei Buchstaben (siehe Tabelle 3).

Regel	Zellfarbe	Schriftfarbe
=K5="L"	Rot	Weiß
=K5="K"	Gelb	Grün
=K5="J"	Grün	Weiß

Tabelle 3: Für die Buchstaben J, K und L legen Sie Formatierungsregeln fest

Testen Sie nun die Ampelfunktion, schreiben Sie Istkosten und Restkosten in die ersten Zeilen mit Phasenzuweisung. Das Bedingungsformat färbt die Zellen der Spalte K entsprechend ein, wenn die Formel die Übereinstimmung von Soll- und Istkosten berechnet hat (Bild 8).

Der Projektstrukturplan enthält jetzt die Plandaten für die Termine und Kosten der Arbeitspakete oder – wie in unserem Beispiel – der Phasen.




Bild 8: Kostenvisualisierung mit Ampelformatierung

Berichtswesen nach Projektstart aufsetzen

Lassen Sie uns nun das Projekt starten und Ist-Daten erfassen. Dazu hat der Projektleiter oder das Projektteam ein regelmäßiges Berichtswesen beschlossen. Per E-Mail oder über Intranet-Formulare werden von den einzelnen Teammitgliedern Statusberichte angefordert, diese melden periodisch alle terminlichen Änderungen und die angefallenen Istkosten. Als Berichtszeitpunkt wird die Kalenderwoche vereinbart.

Istkosten als Szenarien speichern

Szenarien sind eine nützliche Technik, um den Status einer Tabelle oder eines Tabellenbereichs zu sichern. Sie erstellen dabei sozusagen eine Momentaufnahme der jeweiligen Werte, die Sie zu einem beliebigen späteren Zeitpunkt wieder abrufen können. In unserem Beispiel verwenden wir Szenarien, um die Entwicklung der Istkosten zu überwachen. Dazu speichern Sie die entsprechenden Werte zu jedem Berichtszeitpunkt in einem Szenario. Anschließend können Sie die verschiedenen Szenarien miteinander vergleichen, um Trends festzustellen. Wie Sie aus den gespeicherten Szenarien eine Kostentrend-Analyse erstellen können, zeigt der zweite Teil des Beitrags.



Achten Sie darauf, dass Szenarien nur Texte oder Zahlen speichern können, über Formeln berechnete Werte sind nicht erlaubt. Mit der Beschränkung auf maximal 32 veränderbare Zellen eignet sich die Technik auch nicht, um ganze Projektpläne zu speichern.

Das Projekt ist gestartet, nach der ersten Woche werden die Istkosten erfasst und wie folgt in ein erstes Szenario übertragen:

- Tragen Sie die Istkosten und die Restkosten der ersten Phase in Zelle H5 bzw. I5 ein.
- Reduzieren Sie die Ansicht auf die Zeilengliederungsebene 1. Klicken Sie dazu auf das Symbol links oben in der Reihe der Gliederungssymbole.
- Markieren Sie die Zellen der Spalten H und I (H5:I34), und drücken Sie F5 (Start / Bearbeiten / Gehe zu). Wählen Sie Inhalte / Nur sichtbare Zellen.
- Wählen Sie Daten / Datentools / Was wäre wenn-Analyse / Szenario-Manager. Der Szenario-Manager wird aktiv. Klicken Sie auf Hinzufügen... und geben Sie dem ersten Szenario den Namen "Kosten Designphase".
- Die veränderbaren Zellen werden angezeigt, klicken Sie auf OK, und bestätigen Sie die Inhalte noch einmal mit OK.
- Schließen Sie den Szenario-Manager mit Klick auf die Schaltfläche Schließen.

Geben Sie in einem nächsten Schritt neue Kosten für die nächste Kalenderwoche ein, und erstellen Sie erneut ein Szenario mit den neuen Daten. Im Szenario-Manager können Sie die Werte eines Szenarios auch abändern und so einen bereits gespeicherten Kostenstand korrigieren.

Nutzen Sie auch das Szenarien-Symbol, um Szenarien schnell wechseln oder ändern zu können (Bild 9):

- Wählen Sie Symbolleiste für den Schnellzugriff anpassen im Kontextmenü der kleinen Symbolleiste.
- Schalten Sie um auf die Ansicht Alle Befehle, suchen Sie den Eintrag Szenario und holen Sie ihn per Klick auf Hinzufügen in die Symbolleiste. Bestätigen Sie mit OK.

Jetzt können Sie die Szenarien über das Symbol in der Symbolleiste für den Schnellzugriff anzeigen und wechseln (Bild 9).




Bild 9: Gespeicherte Szenarien werden über ein Symbol in der kleinen Symbolleiste abgerufen

Kostensummen berechnen

Die Summen der einzelnen Kosten (Plan-, Ist-, Soll- und Restkosten) berechnen Sie außerhalb des Szenarien-Bereichs, am besten unterhalb des Projektplans. Geben Sie Folgendes ein:

Zelle F40: "Summe:"

Zelle G40: =SUMME(G4:G38)

Kopieren Sie die Formel per Füllkästchen nach rechts bis zu den Sollkosten (J40). Jetzt sehen Sie die Kostensummen auch beim Umschalten auf die einzelnen Szenarien (Bild 10).

ID	Vorgang	Beginn	Dauer	Ende	Projekttag	Plankosten	Istkosten	Restkosten	Sollkosten	Ampel
1	Projekt "Magic AirStar"									
4	Projektbeginn			Di 03.01.17						
5	Design					18.300	18.300	2.000	20.300	
6	Produktplanungsbesprechung durchführen	Di 03.01.17	12 t	So 15.01.17	8 t					
7	Skizzen erstellen	Di 10.01.17	7 t	Di 17.01.17	6 t					
8	Produktdesign festlegen	Di 17.01.17	9 t	Do 26.01.17	8 t					
9	Marketingstrategie festlegen	Do 26.01.17	11 t	Mo 06.02.17	8 t					
10	Technische Zeichnungen erstellen	Mo 06.02.17	10 t	Do 16.02.17	9 t					
11	Abschluss Designphase			Do 16.02.17						
12	Gehäuse					33.000	26.000	3.000	29.000	
13	Material auswählen	Do 16.02.17	5 t	Di 21.02.17	4 t					
14	Gehäuse zeichnen	Mi 22.02.17	12 t	Mo 06.03.17	9 t					
15	Gittermodell erstellen	Mo 06.03.17	4 t	Fr 10.03.17	5 t					
16	Muster herstellen	Mo 13.03.17	8 t	Di 21.03.17	7 t					
17	Abnahme Gehäuse			Di 21.03.17						
18	Elektronik					48.000	41.000	10.000	51.000	
19	Materialbeschaffung	Di 28.02.17	6 t	Mo 06.03.17	5 t					
20	Leiterplatten entwickeln	Mo 06.03.17	14 t	Mo 20.03.17	11 t					
21	Elektronik fertigen	So 12.03.17	14 t	So 26.03.17	10 t					
22	Abnahme Elektronik			So 26.03.17						
23	Motor					35.000	29.000	5.000	34.000	
24	Lieferantenauswahl	Do 16.02.17	5 t	Di 21.02.17	4 t					
25	Motor entwickeln	Di 21.02.17	13 t	Mo 06.03.17	10 t					
26	Motor fertigen	Mo 06.03.17	21 t	Mo 27.03.17	16 t					
27	Abnahme Motor			Mo 27.03.17						
28	Testphase					15.000	16.000	1.000	17.000	
29	Funktionsmuster herstellen	Mo 03.04.17	4 t	Fr 07.04.17	5 t					
30	Funktionsmuster testen	Mo 10.04.17	4 t	Fr 14.04.17	4 t					
31	Werkzeug vorbereiten	So 16.04.17	3 t	Mi 19.04.17	2 t					
32	Nacharbeit	Mi 19.04.17	5 t	Mo 24.04.17	4 t					
33	Abschluss Testphase			Mo 24.04.17						
34	Nullserienfertigung					62.000	45.000	10.000	55.000	
35	Nullserienbesprechung	Mi 26.04.17	2 t	Fr 28.04.17	3 t					
36	Nullserie produzieren	Mo 01.05.17	12 t	Sa 13.05.17	9 t					
37	Nullserie fertiggestellt			Sa 13.05.17						
38	Projektende			Sa 13.05.17						
39										
40	Summe:					211.300	175.300	31.000	206.300	

Bild 10: Die Kostensummen werden unter dem Projektplan ermittelt

Fazit und Ausblick

Im ersten Teil dieses Beitrags haben Sie aus einer Auflistung von Projektaktivitäten einen funktionellen Projektplan mit Phasen, Meilensteinen und Vorgangsdauerberechnung erstellt. Mit Hilfe von Gliederungsfunktion und Autofilter lässt sich dieser in einen Phasen- oder Meilensteinplan umwandeln. Die Bedingungsformatierung sorgt für eine nützliche Optik mit Farbkennzeichnung und Ampelformatierung. In weiteren Spalten des Projektplans verwalten Sie die Plan- und Istkosten, eine Ampelformatierung sorgt hierbei für die passende Optik. Szenarien bieten die Möglichkeit, die Kostenentwicklung in den einzelnen Phasen zu archivieren.

Der zweite Teil des Beitrags zeigt, wie Sie zur Überwachung der Termine und Kosten aus den Daten des Projektplans mit Hilfe von PivotTable-Berichten und PivotCharts eine Meilensteintrend- sowie eine Kostentrend-Analyse erstellen.

Literatur

Schels, Ignatz: Projektmanagement mit Excel, Hanser-Verlag, München, <https://www.projektmagazin.de/projektmanagement-buecher/projektmanagement-mit-excel-1097347>